Sending Outlook Appointments and Meeting Requests Directly from ABAP Applications

Applies to:
SAP Net Weaver 7.0, ABAP. For more information, visit the ABAP homepage.

Summary
This sample code demonstrates how to send Appointments or Meetings Requests to e.g. Microsoft Outlook directly from a NetWeaver (6.20 or higher) based system. There are many possible ways to use this, one very logical and helpful use is to automatically send “vacation”/Leave of absence appointments back to a respective employee and his manager after vacation/leave request approval in SAP HCM.

Author: Martin Lang
Company: SAP Americas Inc.
Created on: July 30, 2009

Author Bio
Martin Lang is the Director for Business Process Integration with SAP IT, working from Newtown Square, PA, USA since 7 years. With a team of dedicated professionals he pursues the mission of integrating and improving end-to-end processes.
Table of Contents

Summary............................................................................................................................................................. 3
  Pre-requisites .................................................................................................................................................. 3
  Overview and Result ....................................................................................................................................... 3
  Sample Code .................................................................................................................................................. 4
Related Content.................................................................................................................................................. 7
Copyright............................................................................................................................................................. 8
Summary

Sending eMails from an SAP NetWeaver based system is a very basic Standard functionality. Sending appointments/meeting requests however, also directly from SAP NetWeaver is something more unique, but it's supported nonetheless and can be very useful e.g. for

- Automatically Sending “vacation”/Leave of absence appointments back to the employee and his manager after vacation/leave request approval
- Automatically sending resubmission/follow-up appointments e.g. x days before contract expiration
- ...  

Pre-requisites

Sending out Appointments to e.g. Microsoft Outlook essentially makes use of the ABAP Class CL_APPOINTMENT.

The Class should exist and work on systems SAP NetWeaver 6.20 and higher.

However, you need to make sure that these four notes are applied or obsolete based on a higher release/SP level:

<table>
<thead>
<tr>
<th>Number</th>
<th>Short Text of Note</th>
</tr>
</thead>
<tbody>
<tr>
<td>1364539</td>
<td>Incorrect display of ICS documents</td>
</tr>
<tr>
<td>1353329</td>
<td>iCalendar document that is sent is processed incorrectly</td>
</tr>
<tr>
<td>1357483</td>
<td>Title of sent appointment is incorrect</td>
</tr>
<tr>
<td>1369864</td>
<td>All-Day Appointment doesn’t appear correctly in Outlook</td>
</tr>
</tbody>
</table>

Note: Implement these notes with transaction SNOTE, which automatically checks for the applicability based on your current versions/patchlevels

Overview and Result

The program below is a sample to demonstrate how appointments can be sent touching on many of the available options like:

- multiple attendees
- Define Appointment Subject, Location & Body Long Text
- Specify Start/End Date/Time
- Create All-Day-Appointments
- Define “Show as”: Free,Busy etc.
This screenshot represents an appointment sent through the sample code below and received with Microsoft Outlook 2007:

### Sample Code

**Sample code for demonstration purposes:**

```abap
*---------------------------------------------------------------------*
* Report  z_appointment_to_outlook.                                *
*---------------------------------------------------------------------*

REPORT z_appointment_to_outlook.

INCLUDE <cntn01>.

TYPE-Pools: SCCON.

DATA: MAIL TYPE TEXT40.

* Provide select option to enter multiple attendees
SELECTION-SCREEN: BEGIN OF BLOCK org.
SELECT-OPTIONS: EMAIL FOR MAIL NO INTERVALS DEFAULT 'Joe.Doe@test.com'.
SELECTION-SCREEN: END OF BLOCK org.

DATA: LO_APPONMENT TYPE REF TO CL_APPOINTMENT.
DATA: LS_PARTICIPANT TYPE SCSPART.
DATA: LV_ADDRESS TYPE SWC_OBJECT.
DATA: LS_ADDRESS_CONTAINER LIKE SWCONT OCCURS 0 WITH HEADER LINE.
DATA: LT_TEXT TYPE SO_TXTTAB.
DATA: LS_TEXT LIKE LINE OF LT_TEXT.
DATA: LV_LOCATION LIKE LINE OF LS_TEXT.
```

SAP COMMUNITY NETWORK - sdn.sap.com | BPX - bpx.sap.com | BOC - boc.sap.com

© 2009 SAP AG
data lo_send_request type ref to cl_bcs.
data lv_sent_to_all type os_boolean.

create object lo_appointment.

* Add multiple attendees
loop at email.
clear ls_participant.
swc_create_object lv_address 'ADDRESS' space.
swc_set_element ls_address_container 'AddressString' email-low.
swc_set_element ls_address_container 'TypeId' 'U'.
swc_call_method lv_address 'Create' ls_address_container.
check sy-subrc = 0.

* Get key and type of object
swc_get_object_key lv_address ls_participant-objkey.
check sy-subrc = 0.
swc_get_object_type lv_address ls_participant-objtype.
check sy-subrc = 0.
move sccon_part_sndmail_with_ans to ls_participant-send_mail.
ls_participant-comm_mode = 'INT'.
lo_appointment->add_participant( participant = ls_participant ).
endloop.

* Sample Appointment for specific date/time
lo_appointment->set_date( date_from = '20090801'
                          time_from = '130000'
                          date_to = '20090801'
                          time_to = '140000'
                          ).

* Sample Appointment for All-Day-Event
lo_appointment->set_date( date_from = '20090803'
                          date_to = '20090803'
                          ).

* Make appointment appear "busy"
lo_appointment->set_busy_value( sccon_busy_busy ).

* Set Location
lo_appointment->set_location_string( 'Location' ).

* Set Organizer
lo_appointment->set_organizer( sy-uname ).

* "Type of Meeting" (value picked from table SCAPPTTYPE)
lo_appointment->set_type( 'ABSENT' ).

* Make this an all day event
lo_appointment->SET_VIEW_ATTRIBUTES( SHOW_ON_TOP = 'X' ).

* Set Meeting body text
ls_text = 'This is the Body Text of the Appointment'.
append ls_text to lt_text.
lo_appointment->set_text( lt_text ).

* Set Meeting Subject
lo_appointment->set_title( 'This is the Appointment Subject' ).

* Important to set this one to space. Otherwise SAP will send a not user-friendly e-mail
lo_appointment->save( send_invitation = space ).

* Now that we have the appointment, we can send a good one for outlook by switching to BCS
lo_send_request = lo_appointment->create_send_request( ).

* don't request read/delivery receipts
lo_send_request->set_status_attributes( i_requested_status = 'N' i_status_mail = 'N' ).

* Send it to the world
lv_sent_to_all = lo_send_request->send( i_with_error_screen = 'X' ).

commit work and wait.
Related Content
For more information, visit the ABAP homepage.